
program mhd; 
{

Uloha celoslovenskeho kola sutaze Zenit v programovani 
             Piestany 29.11.1990 

1. Listok MHD ma tvar :      55   - cislo listka 
                            . . . 
                            . . . 
                            . . . 
   jednotlive bodky sa mozu predierkovat. 

   Napiste program, ktory pre dane cislo  k  ( 0 < k < 10 ) 
   vypise vsetky rozlicne listky mestskej hromadnej dopravy 
   s k dierkami a ocisluje ich. Pritom setri papierom a vypisuje 
   m listkov do jedneho radu vedla seba. Dva listky su rozne, ak 
   sa nedaju stotoznit ani otocenim okolo zvislej osy. 

   } 
   uses crt; 
   type    listok = string[9]; 
   const   m      = 15; 
           znaky_riadku   = 80; 
           me = ''; mev = '  ';  des = 3; 
           cisty  : listok = '.........'; 
           koniec : boolean = false; 
           dierka : char = 'o'; 
           bodka  : char = '.'; 
var
           r      : array[1..m] of listok; 
           pocet,pocet_v_riadku : integer; 
           t      : text; 
           d      : array[1..9] of integer; 
           l      : listok; 
           k      : integer; 

function este_nebol:boolean; 
var sym : listok; 
begin
  if not koniec then 
    begin 
      sym:=cisty; 
      sym[1]:=l[3];  sym[3]:=l[1];  sym[2]:=l[2]; 
      sym[4]:=l[6];  sym[6]:=l[4];  sym[5]:=l[5]; 
      sym[7]:=l[9];  sym[9]:=l[7];  sym[8]:=l[8]; 
      este_nebol:=l>=sym; 
    end 
  else este_nebol:=false; 
end;

procedure priprava; 
var  i : integer; 
begin
  pocet:=1;  pocet_v_riadku:=1; 
  assign(t,'t.t');  rewrite(t); 
  repeat 
    clrscr; 
    write('kolko dierkove listky ? : ');  readln(k); 
  until (k>0) and (k<9); 
  l :=cisty; 
  for i:=1 to k do begin d[i]:=i; l[i]:=dierka; end; 
  r[1]:=l; 
  writeln(t); 
  writeln(t,'MHD pocet dierok = ',k); 
end;



procedure generuj; 
var i,pocet_dierok : integer; 
begin
  if l[9]=dierka then 
    begin 
      i:=9; 
      repeat l[i]:=bodka; 
             i:=i-1; 
      until  l[i]=bodka; 
      pocet_dierok:=9-i; 
      koniec:=(pocet_dierok=k); 
      if not koniec then 
        begin 
          repeat i:=i-1  until l[i]=dierka; 
          l[i]:=bodka;  l[i+1]:=dierka; 
          for i:=i+2 to i+1+pocet_dierok do l[i]:=dierka; 
        end; 
    end 
  else 
    begin 
      i:=9; repeat i:=i-1  until l[i]=dierka; 
      l[i]:=bodka;  l[i+1]:=dierka; 
    end; 
  if este_nebol then 
    begin 
      pocet:=pocet+1; inc(pocet_v_riadku); 
      r[pocet_v_riadku]:=l; 
    end; 
end;

procedure tlac_riadok(kolko:integer); 
var i,j : integer; 
begin
  writeln(t); 
  for i:=1 to kolko do begin 
    write(t,pocet-kolko+i:des,mev); 
  end; 
  writeln(t); 
  for i:=1 to 3 do begin 
    for j:=1 to kolko do begin 
      write(t,r[j,i*3-2],r[j,3*i-1],me,r[j,i*3],me,mev); 
    end; 
    writeln(t); 
  end; 
  writeln(t); 
end;

begin
  priprava; 
  repeat 
    generuj; 
    if pocet_v_riadku = m then 
      begin tlac_riadok(m); pocet_v_riadku:=0; 
      end; 
  until koniec; 
  if pocet_v_riadku>0 then tlac_riadok(pocet_v_riadku); 
  writeln(t); 
  writeln(t,'pocet roznych listkov : ',pocet); 
  close(t); 
  repeat until readkey = #27; 
end.



program mnozina; 
{

Uloha celoslovenskeho kola sutaze Zenit v programovani 
             Piestany 29.11.1990 

             2. Mnozina M je dana nasledovne : 
                1 . cislo 1 patri do M 
                2. ak patri x do M, potom tam patria aj cisla 2x+1,3x+1,5x. 
                3. ziadne ine cisla do M nepatria 
                Vypiste pre dane n ( n <= 1000 ) n najmensich prvkov mnoziny 
                vzostupne usporiadanych. 

}
uses crt; 
var
    a             : array[1..1003] of integer; 
    n,p2,p3,p5,p  : integer; 
    i,c2,c3,c5    : integer; 
    t             : text; 
begin
  assign(t,'t.t'); rewrite(t); 
  clrscr; 
  write('n : '); readln(n); 
  p:=1; a[1]:=1; 
  p2:=1; p3:=1; p5:=1; 
  c2:=3; c3:=4; c5:=5; 
  repeat 
      if (c2<=c3) and (c2<=c5) then  {posun pointer  p2 } 
        begin 
          if a[p]<c2 then begin p:=p+1; a[p]:=c2;  end; 
          inc(p2); c2:=2*a[p2]+1; 
        end; 
      if (c3<=c2) and (c3<=c5) then   { posun pointer  p3 } 
        begin 
          if a[p]<c3 then begin p:=p+1; a[p]:=c3; end; 
          inc(p3); c3:=3*a[p3]+1; 
        end; 
      if (c5<=c2) and (c5<=c3) then  {posun pointer  p5} 
        begin 
          if a[p]<c5 then begin p:=p+1; a[p]:=c5; end; 
          inc(p5); c5:=5*a[p5]; 
        end; 
  until p>=n; 
  writeln(t); 
  writeln(t,'Mnozina  n : ',n); 
  for i:=1 to n do 
    begin 
      if i mod 10 = 1 then begin writeln(t); write(t,i:3,'.'); end; 
      write(t,a[i]:6); 
    end; 
  writeln(t); 
  close(t); 
end.


